
Option Explicit  'makes sure that no variables are misspelled.
'Even if public variables are declared to save having to declare them in
'all subroutines, their values do not transfer from one subroutine to
'another.  Such transfers must be done using formal parameters in
'subroutine calls, or through text boxes on the userform.

'---------------------------
Private Sub PREPARE_CommandButton_Click()
'Even if public variables are declared to save having to declare them in
'all subroutines, their values do not transfer from one subroutine to
'another.  That function must be done with formal parameters in
'subroutine calls, or in screen text boxes.

'Declare the variables
Dim refcell As String, colrow As String, blank As String
Dim datastring As String, colon As String, comma As String
Dim top_left As String, bottom_right As String
Dim top_left_copy As String, bottom_right_copy As String
Dim color_light_yellow As String, color_light_green As String
Dim color_light_red As String, color_white
Dim ia_maxdim As Integer, ia_dim As Integer
Dim jrow As Integer, jcol As Integer, jrow_last As Integer
Dim krow As Integer, kcol As Integer
Dim error_tolerance As Single

'Initialize variables
blank = ""
colon = ":"
comma = ","
color_light_yellow = 13434879  'on spreadsheet
color_light_green = 13434828   'on spreadsheet
color_light_red = 13421823     'on spreadsheet
'refcell_textbox.Text = blank
ia_dim_textbox.Text = blank
'ia_maxdim_textbox.Text = blank

Prepare_Textbox.BackColor = &H80000005  'White, on userform
Total_Abs_Error_Textbox.Text = blank
Total_Abs_Error_Textbox.BackColor = &H80000005  'White, on userform

'refcell = "A5"  '1 column left, and 1 row above the A matrix
refcell = refcell_textbox.Text
top_left = "A1"
bottom_right = "V39"  'the entire working area
'ia_maxdim = 15
ia_maxdim = CInt(ia_maxdim_textbox.Value)
top_left_copy = "A25"  'the A,x,b copy areas
bottom_right_copy = "V39"

colrow = "A1"  'Fill in name and description from the userform
Range(colrow).Select
ActiveCell.Value = Gaussian_Elimination_Excel.Caption & ", " & Version_Textbox.Text
ActiveCell.Offset(1, 0).Value = Name_Textbox.Text
ActiveCell.Offset(2, 0).Value = Description_Textbox.Text

'Clear backcolors from the working area
colrow = refcell
Range(top_left & colon & bottom_right).Interior.Pattern = xlNone

'Clear contents of the copy areas below Ax=b
Range(top_left_copy & colon & bottom_right_copy).ClearContents

'Set the reference cell for the solution process
Range(colrow).Select

'Clear the error box
ActiveCell.Offset(0, ia_maxdim + 6).ClearContents
ActiveCell.Offset(0, ia_maxdim + 6).Interior.Pattern = xlNone

jrow_last = 0
For jrow = 1 To ia_maxdim  'Color light green the portions of A, x, b to be used
  jcol = jrow
  datastring = ActiveCell.Offset(jrow, jcol)
  If datastring <> blank Then
    ActiveCell.Offset(jrow, jcol).Interior.Color = color_light_green 'For A diag.
    If jrow > 1 Then
      For krow = 1 To jrow - 1 'For A rows above diag, and A cols that are left of diag.
        ActiveCell.Offset(krow, jcol).Interior.Color = color_light_green 'for rows above
        ActiveCell.Offset(jcol, krow).Interior.Color = color_light_green 'for cols on the left
      Next krow
    End If
    ActiveCell.Offset(jrow, ia_maxdim + 4).Interior.Color = color_light_green  'for the row of b
    jrow_last = jrow
    Else
    GoTo exit_1
  End If
Next jrow
exit_1:
Prepare_Textbox.BackColor = &HFF00&      'Green, on userform
SOLVE_CommandButton.Enabled = True
If jrow_last = 0 Then
  ActiveCell.Interior.Color = color_light_red
  Prepare_Textbox.BackColor = &HFF&      'Red, on userform
  SOLVE_CommandButton.Enabled = False
  GoTo finished
End If
ia_dim = jrow - 1 'dimension of the problem
'convert blanks to zero in the problem area
For jrow = 1 To ia_dim
  For jcol = 1 To ia_dim
    If jrow <> jcol Then 'For A
      If ActiveCell.Offset(jrow, jcol).Value = blank Then
        ActiveCell.Offset(jrow, jcol).Value = 0
      End If
    End If
  Next jcol
  ActiveCell.Offset(jrow, ia_maxdim + 2).Value = blank 'For x
 'End If
  If ActiveCell.Offset(jrow, ia_maxdim + 4).Value = blank Then
    ActiveCell.Offset(jrow, ia_maxdim + 4).Value = 0 'For b
  End If
Next jrow

'refcell_textbox.Text = refcell
ia_dim_textbox.Value = ia_dim
'ia_maxdim_textbox.Value = ia_maxdim

On Error GoTo tol_err
GoTo finished
tol_err:  'picks up error when reading format error
  error_tolerance = CSng(Error_Tolerance_Textbox.Value)
  Error_Tolerance_Textbox.BackColor = &H80000005  'White

finished:

End Sub

'---------------------------
Private Sub SOLVE_CommandButton_Click()
Dim refcell As String, colrow As String
Dim ia_dim As Integer, ia_maxdim As Integer, jpivot As Integer
Dim jrow As Integer, jcol As Integer, jpv As Integer
Dim pivot As Double, alpha As Double
Dim xvect(15) As Double, b(15) As Double, bcheck(15) As Double
Dim color_light_yellow As String, color_light_green As String
Dim color_light_red As String, graph_title As String
Dim total_abs_error As Single, error_tolerance As Single

color_light_yellow = 13434879
color_light_green = 13434828
color_light_red = 13421823
total_abs_error = 0#

'Prevent going backward without pressing the Prepare commandbutton again
Prepare_Textbox.BackColor = &H80000005  'White
SOLVE_CommandButton.Enabled = False

'The solution process
refcell = refcell_textbox.Text
ia_dim = ia_dim_textbox.Value
ia_maxdim = ia_maxdim_textbox.Value
colrow = refcell
Range(colrow).Select

'Make a copy of the A matrix and b vector below the originals
For jrow = 1 To ia_dim
  For jcol = 1 To ia_dim
    ActiveCell.Offset(jrow + ia_maxdim + 4, jcol).Value = ActiveCell.Offset(jrow, jcol).Value
  Next jcol
  b(jrow) = ActiveCell.Offset(jrow, ia_maxdim + 4).Value
  ActiveCell.Offset(jrow + ia_maxdim + 4, ia_maxdim + 4).Value = b(jrow)
Next jrow

On Error GoTo Divide_by_zero
'Gaussian Elimination on the A matrix and the b vector.  When finished,
'the A matrix is upper-right triangle plus diagonal form, URT
For jpivot = 1 To ia_dim - 1
'For matrix A, divide the pivot row by the pivot value
  pivot = ActiveCell.Offset(jpivot, jpivot).Value
  For jcol = jpivot To ia_dim
    ActiveCell.Offset(jpivot, jcol).Value = ActiveCell.Offset(jpivot, jcol).Value / pivot
  Next jcol
'Also for vector b
  ActiveCell.Offset(jpivot, ia_maxdim + 4).Value = ActiveCell.Offset(jpivot, ia_maxdim + 4).Value / pivot
'For matrix A, subtract alpha * pivot row from rows below the pivot row
  For jrow = jpivot + 1 To ia_dim
    alpha = ActiveCell.Offset(jrow, jpivot)
    For jcol = jpivot To ia_dim
      ActiveCell.Offset(jrow, jcol) = ActiveCell.Offset(jrow, jcol) - alpha * ActiveCell.Offset(jpivot, jcol)
    Next jcol
'Also for vector b
    ActiveCell.Offset(jrow, ia_maxdim + 4).Value = ActiveCell.Offset(jrow, ia_maxdim + 4) - alpha * ActiveCell.Offset(jpivot, ia_maxdim + 4)
  Next jrow
Next jpivot

'Backward Substitution to solve for vector x
jpivot = ia_dim
pivot = ActiveCell.Offset(jpivot, jpivot).Value
xvect(jpivot) = ActiveCell.Offset(jpivot, ia_maxdim + 4).Value / pivot
If ia_dim > 1 Then
  For jpv = 1 To ia_dim - 1
    jpivot = ia_dim - jpv  'jpivot begins with Row (iadim - 1), going backward to Row 1
    pivot = ActiveCell.Offset(jpivot, jpivot).Value
    xvect(jpivot) = ActiveCell.Offset(jpivot, ia_maxdim + 4)
    For jcol = jpivot + 1 To ia_dim
      xvect(jpivot) = xvect(jpivot) - ActiveCell.Offset(jpivot, jcol) * xvect(jcol)
    Next jcol
    xvect(jpivot) = xvect(jpivot) / pivot
  Next jpv
End If

'write vector x into the spreadsheet cells
For jpivot = 1 To ia_dim
  ActiveCell.Offset(jpivot, ia_maxdim + 2).Value = xvect(jpivot)
  ActiveCell.Offset(ia_maxdim + 4 + jpivot, ia_maxdim + 2).Value = xvect(jpivot)
Next jpivot

'Check solution Ax = b.
For jrow = 1 To ia_dim
  bcheck(jrow) = 0
  For jcol = 1 To ia_dim
    bcheck(jrow) = bcheck(jrow) + ActiveCell.Offset(ia_maxdim + 4 + jrow, jcol).Value * xvect(jcol)
  Next jcol
  ActiveCell.Offset(ia_maxdim + 4 + jrow, ia_maxdim + 6).Value = bcheck(jrow)  'Write b
  total_abs_error = total_abs_error + Abs(bcheck(jrow) - b(jrow))
Next jrow

'Total Abs Error
'total_abs_error = 3.1415926  'use this for testing purposes
ActiveCell.Offset(0, ia_maxdim + 6).Value = total_abs_error
Total_Abs_Error_Textbox.Text = Format(total_abs_error, "0.00E+00")
error_tolerance = Error_Tolerance_Textbox.Value
If total_abs_error > error_tolerance Then  'color the error alerts red
  Total_Abs_Error_Textbox.BackColor = &HFF&    'Red, on userform
  ActiveCell.Offset(0, ia_maxdim + 6).Interior.Color = color_light_red    'total error cell
  Else
  Total_Abs_Error_Textbox.BackColor = &HFF00&  'Green, on userform
  ActiveCell.Offset(0, ia_maxdim + 6).Interior.Color = color_light_green  'total error cell
End If

On Error GoTo finished

colrow = "BK5"   'Location of Correlation Coefficient R, then Polynomial Order
Range(colrow).Select
ActiveCell.Offset(0, 1) = ia_dim_textbox.Value - 1
ActiveSheet.ChartObjects("Chart 1").Activate
ActiveChart.ChartTitle.Select
graph_title = "Polynomial Order = " & ActiveCell.Offset(0, 1)
graph_title = graph_title & ", R = " & Format(ActiveCell, "0.000")
ActiveChart.ChartTitle.Text = graph_title
'Selection.Format.TextFrame2.TextRange.Characters.Text = "mack"

GoTo finished

Divide_by_zero:
ActiveCell.Offset(jpivot, jpivot).Interior.Color = color_light_red  'divide by pivot gave error

finished:

colrow = "AZ2"  'min x for graph            'Specific Cell Location
Range(colrow).Select
If Curve_Fit_Checkbox.Value = True Then
  ActiveCell.Offset(0, 0) = "=MIN(X6:X600)" 'Specific Cell Location
  ActiveCell.Offset(1, 0) = "=MAX(X6:X600)" 'Specific Cell Location
  Else
  ActiveCell.Offset(0, 0) = 1.01
  ActiveCell.Offset(1, 0) = -0.99
  ActiveSheet.ChartObjects("Chart 1").Activate
  ActiveChart.ChartTitle.Select
  graph_title = "Not Created"
  ActiveChart.ChartTitle.Text = graph_title
End If
Curve_Fit_Checkbox.Value = False
colrow = refcell
Range(colrow).Select
End Sub

'---------------------------
Private Sub Random_Load_And_Solve_Commandbutton_Click()
Dim refcell As String, colrow As String
Dim ia_maxdim As Integer, ia_dim As Integer
Dim jrow As Integer, jcol As Integer

refcell = refcell_textbox.Text
ia_maxdim = ia_maxdim_textbox.Value
colrow = refcell
Range(colrow).Select

Randomize  'starts the random number generator at a different point each time Randomize is called

For jrow = 1 To ia_maxdim
  For jcol = 1 To ia_maxdim  'the A matrix
   'Rnd yields 0 to 1
    ActiveCell.Offset(jrow, jcol).Value = 200# * Rnd() - 100  'This way we get -100 to +100
  Next jcol
  ActiveCell.Offset(jrow, ia_maxdim + 4).Value = 200# * Rnd() - 100   'the b vector
Next jrow

PREPARE_CommandButton_Click
SOLVE_CommandButton_Click

Range(colrow).Select

End Sub

'---------------------------
Private Sub DO_10_Commandbutton_Click()
Dim refcell As String, colrow As String
Dim ia_maxdim As Integer, ia_dim As Integer
Dim jrow As Integer, jcol As Integer

refcell = refcell_textbox.Text
ia_maxdim = ia_maxdim_textbox.Value
colrow = refcell
Range(colrow).Select

'For jrow = 1 To 10
'  ActiveCell.Offset(jrow - 1, ia_maxdim + 10).ClearContents
'Next jrow

For jrow = 1 To 10
  Random_Load_And_Solve_Commandbutton_Click
'  ActiveCell.Offset(jrow - 1, ia_maxdim + 10).Value = Total_Abs_Error_Textbox.Value
  If Total_Abs_Error_Textbox.BackColor = &HFF& Then GoTo got_error
Next jrow
got_error:
End Sub

